



ESKİŞEHİR OSMANGAZİ ÜNİVERSİTESİ

Mühendislik Mimarlık Fakültesi

İnşaat Mühendisliği Bölümü

E-Posta: ogu.ahmet.topcu@gmail.com

Web: <http://mmf2.ogu.edu.tr/atopcu>

Bilgisayar Destekli Nümerik Analiz

Ders notları 2014

Ahmet TOPÇU

```
READ n, m, L: ' DATA satirlerinden oku
DIM a(n, m), b(m, L), c(n, L): ' Matrislere yer ayır
CLS : ' ekranı temizle
CALL Matrisoku(n, m, a())
CALL Matrisoku(m, L, b())
CALL matriscarp(n, m, L, a(), b(), c())
PRINT "A:"
CALL matrisyaz(n, m, a())
PRINT "B:"
CALL matrisyaz(m, L, b())
PRINT "C:"
CALL matrisyaz(n, L, c())
CALL Determinant(n, c(), rank, d, k, v)
PRINT "r="; rank, "d="; d; "k="; k; "v="; v
END
```

```
DEFDBL A-H, O-Z
SUB matriscarp (n, m, L, a(), b(), c())
' C(n,L)=A(n,m)*B(m,L) hesaplanır
-----
FOR i = 1 TO n
  FOR j = 1 TO L
    T = 0
    FOR k = 1 TO m
      T = T + a(i, k) * b(k, j)
    NEXT k
    c(i, j) = T
  NEXT j
NEXT i
END SUB
```

12

PROGRAMLAR: Basit matris işlemleri

12. PROGRAMLAR: Basit matris işlemleri

```

' Matris işlemleri ana programı
' Matrislerin boyutları: n,m,l
DATA 2,3,2
'A n'ın elemanları:
DATA 1,2,3
DATA 4,5,6
'B n'ın elemanları:
DATA 1,2
DATA 2,0
DATA 3,1

DEFINT I-N: 'I,J,K,L,M,N harfleri ile başlayan değişkenler tam sayıdır.
DEFDBL A-H, O-Z: 'Bu harflerle başlayanlar çift hassasiyetli ondalık sayıdır
DECLARE SUB Matrisoku(n, m, a())
DECLARE SUB matrisokuSimetrik(n, a())
DECLARE SUB matrisyaz(n, m, a())
DECLARE SUB vektoroku(n, v())
DECLARE SUB vektoryaz(n, v())
DECLARE SUB matriscarp(n, m, L, a(), b(), c())
DECLARE SUB MatrisTopla(Topla, n, m, a(), b(), c())
DECLARE SUB diyagmatriscarp(n, m, d(), a())
DECLARE SUB matrisdiyagcarp(n, m, a(), d())
DECLARE SUB matrisvektorcarp(n, m, a(), v(), c())
DECLARE SUB matriscopyala(n, m, a(), b())
DECLARE SUB transpoz(n, m, a(), b())
DECLARE SUB transpozkare(n, a())
DECLARE SUB birimmatris(n, a())
DECLARE SUB sifirmatris(n, m, a())
DECLARE SUB Determinant(n, a(), rank, d, k, v)

READ n, m, L: 'DATA satırlarından oku

DIM a(n, m), b(m, L), c(n, L): ' Matrislere yer ayır
CLS: ' ekranı temizle
CALL Matrisoku(n, m, a())
CALL Matrisoku(m, L, b())
CALL matriscarp(n, m, L, a(), b(), c())
PRINT "A:"
CALL matrisyaz(n, m, a())
PRINT "B:"
CALL matrisyaz(m, L, b())
PRINT "C:"
CALL matrisyaz(n, L, c())
CALL Determinant(n, c(), rank, d, k, v)
PRINT "r="; rank, "d="; d, "k="; k, "V="; v
END

```

$$A_{2 \times 3} = \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix}$$

$$B_{3 \times 2} = \begin{bmatrix} 1 & 2 \\ 2 & 0 \\ 3 & 1 \end{bmatrix}$$

$C=A \cdot B$ ve $\det C$ program ile hesaplanacak

Hangi harfle başlayan değişkenlerin Tam sayı, hangilerinin ondalık sayı olacağı tanımlanıyor

Alt programların adı ve parametreleri ana programa tanıtılıyor!

n, m, L sayılarını oku DATA satırlarından okur. Bunlar A, B ve C matrislerinin boyutunu tanımlamak için kullanılacak

A, B ve C matrisleri için ana bellekte yer ayır

Matrisoku alt programına git A matrisini oku geri gel

Matrisoku alt programına git B matrisini oku geri gel

Matriscarp alt programına git A ile B yi çarp C ye depola ve geri gel

Matrisyaz alt programına git A yı yaz, gel

Determinant hesaplayan alt programını çağır

Determinant ve kondisyon bilgilerini yaz

```

SUB Matrisoku(n, m, a())
'A(n,m) matrisini Data stırlarından okur
-----
FOR i = 1 TO n
FOR j = 1 TO m
READ a(i, j)
NEXT j
NEXT i
END SUB

```

```

SUB matrisokuSimetrik(n, a())
'A(n,n) simetrik matrisinin üst üçgen
'elemanlarını okur ve alt üçgenini tamamlar
-----
FOR i = 1 TO n
FOR j = i TO n
READ a(i, j)
a(j, i) = a(i, j)
NEXT j
NEXT i
END SUB

```

```

SUB matrisyaz(n, m, a())
'A(n,m) matrisini ekrana yazar
-----
FOR i = 1 TO n
FOR j = 1 TO m
PRINT a(i, j);
NEXT j
PRINT
NEXT i
END SUB

```

Programın sonuçları

```

C:\WINDOWS\system32\cmd.exe - qbasic
A:
1 2 3
4 5 6
B:
1 2
2 0
3 1
C:
14 5
32 14
r= 2          d= .140625 k= 8 U= 519.2494583531118

```

Det $C = d \cdot 2^k = 0.140625 \cdot 2^8 = 36$

$K = \frac{|\det C|}{V} = \frac{36}{519.2} = 0.07$ **C nin HADAMARD kondisyon sayısı**

```

SUB vektoroku (n, v())
' V(n) Vektörünü DATA satırlarından okur
'-----
FOR i = 1 TO n
  READ v(i)
NEXT i
END SUB

```

```

SUB vektoryaz (n, v())
' V(n) Vektörünü ekrana yazar
'-----
FOR i = 1 TO n
  PRINT v(i);
NEXT i
PRINT
END SUB

```

```

SUB matriscarp (n, m, L, a(), b(), c())
' C(n,L)=A(n,m)*B(m,L) hesaplanır
'-----
FOR i = 1 TO n
  FOR j = 1 TO L
    T = 0
    FOR k = 1 TO m
      T = T + a(i, k) * b(k, j)
    NEXT k
    c(i, j) = T
  NEXT j
NEXT i
END SUB

```

```

SUB MatrisTopla (Topla, n, m, a(), b(), c())
'Topla=1 ise C(n,m)=A(n,m)+B(n,m)
'Topla<>1 ise C(n,m)=A(n,m)-B(n,m) hesaplanır
'-----
FOR i = 1 TO n
  FOR j = 1 TO m
    IF Topla = 1 THEN
      c(i, j) = a(i, j) + b(i, j)
    ELSE c(i, j) = a(i, j) - b(i, j)
    END IF
  NEXT j
NEXT i
END SUB

```

```

SUB diyagmatriscarp (n, m, d(), a())
' A(n,m)=D(n)*A(n,m) hesaplanır.
' D(n) vektörü diyagonal matrisin sadece
' diyagonal elemanlarını içerir.
'-----
FOR i = 1 TO n
  FOR j = 1 TO m
    a(i, j) = a(i, j) * d(i)
  NEXT j
NEXT i
END SUB

```

```

SUB matrisediyagcarp (n, m, a(), d())
' A(n,m)=A(n,m)*D(m) hesaplanır.
' D(m) vektörü diyagonal matrisin sadece
' diyagonal elemanlarını içerir.
'-----
FOR i = 1 TO m
  FOR j = 1 TO n
    a(j, i) = a(j, i) * d(i)
  NEXT j
NEXT i
END SUB

```

```

SUB matrisvektorcarp (n, m, a(), v(), c())
' C(n)=A(n,m)*V(m) hesaplanır.
' A(n,m) matrisi V(m) matrisi ile sağdan çarpılır
' C(n) de depolanır.
'-----
FOR i = 1 TO n
  T = 0
  FOR j = 1 TO m
    T = T + a(i, j) * v(j)
  NEXT j
  c(i) = T
NEXT i
END SUB

```

```

SUB matriscopyala (n, m, a(), b())
' A(n,m)=B(n,m) B matrisi A ya kopyalanır
'-----
FOR i = 1 TO n
  FOR j = 1 TO m
    a(i, j) = b(i, j)
  NEXT j
NEXT i
END SUB

```

```

SUB transpoz (n, m, a(), b())
' A(n,m) nin transpozu B(m,n) de depolanır
'-----
FOR i = 1 TO n
  FOR j = 1 TO m
    b(j, i) = a(i, j)
  NEXT j
NEXT i
END SUB

```

```

SUB transpozkatre (n, a())
' A(n,n) kare matrisinin transpozu
' gene A ya depolanır
'-----
FOR i = 1 TO n
  FOR j = i + 1 TO n
    T = a(j, i)
    a(j, i) = a(i, j)
    a(i, j) = T
  NEXT j
NEXT i
END SUB

```

```

SUB birimmatris (n, a())
' A(n,n) matrisini birim matris yapar
'-----
FOR i = 1 TO n
  FOR j = 1 TO n
    a(i, j) = 0
  NEXT j
  a(i, i) = 1
NEXT i
END SUB

```

```

SUB sifirmatris (n, m, a())
' A(n,m) matrisini sıfırlar
'-----
FOR i = 1 TO n
  FOR j = 1 TO m
    a(i, j) = 0
  NEXT j
NEXT i
END SUB

```

```

SUB Determinant (n, a(), rank, d, k, v)
' A(n,n) matrisinin determinanı GAUSS ile hesaplanır
' Dr. Ahmet TOPÇU, Osmangazi Üniversitesi, ESKİŞEHİR, 1994
' A(n,n) çağırın programda depolanmış olmalıdır.
' Programın verdiği d, k ve V sayıları yardımıyla
' det A=d*2^k ile hesaplanabilir.
' A nın kondisyon sayısı Kon A=ABS(det A)/V ile hesaplanabilir
-----
' Norm
rank = 0: v = 1: Zero = 0
FOR i = 1 TO n
  T = 0: d = 0
  FOR j = 1 TO n
    T = T + ABS(a(i, j)): ' WILKINSON normu
    d = d + a(i, j) * a(i, j)
  NEXT j
  IF T > Zero THEN Zero = T
  d = SQR(d): v = v * d: 'HADAMARD hacmi
NEXT i
Zero = Zero * 1E-14

IF n = 1 THEN
' n=1 durumu
IF ABS(a(1, 1)) <= Zero THEN GOTO 100
rank = 1: d = a(1, 1): k = 0
EXIT SUB
END IF

FOR i = 1 TO n - 1
' Kolonda Pivot ara
T = 0: d = 1: k = i
FOR j = i TO n
  IF ABS(a(j, i)) > T THEN T = ABS(a(j, i)): k = j
NEXT j
IF T <= Zero THEN GOTO 100
IF k > i THEN
  d = -d: ' Determinantın işaretini deęiş
' i ve k satırlarını deęiş
FOR j = i TO n
  T = a(i, j)
  a(i, j) = a(k, j)
  a(k, j) = T
NEXT j
END IF

' indirge
rank = i
FOR j = i + 1 TO n
  T = a(j, i) / a(i, i)
  FOR k = i + 1 TO n
    a(j, k) = a(j, k) - T * a(i, k)
  NEXT k
NEXT j
NEXT i
IF ABS(a(n, n)) <= Zero GOTO 100
rank = n

' Determinat için d ve k yi belirle (WILKINSON)
k = 0
FOR i = 1 TO n
  d = d * a(i, i)
200 k = k + 4
  IF ABS(d) >= 1 THEN d = d * .0625: GOTO 200
300 k = k - 4
  IF ABS(d) < .0625 THEN d = d * 16: GOTO 300
NEXT i
EXIT SUB

100 ' Matris tekil
d = 0: k = 0
END SUB

```